<commit_before>{ stdenv, fetchFromGitLab, meson, ninja, pkgconfig
, python36Packages, gnome3, gst_all_1, gtk3, libnotify
, kid3, easytag, gobjectIntrospection, wrapGAppsHook }:

stdenv.mkDerivation rec {
  name = "lollypop-portal-${version}";
  version = "0.9.7";

  src = fetchFromGitLab {
     domain = "gitlab.gnome.org";
     owner = "gnumdk";
     repo = name;
     rev = version;
     sha256 = "0rn5xmh6391i9l69y613pjad3pzdilskr2xjfcir4vpk8wprvph3";
  };

  nativeBuildInputs = [
    gobjectIntrospection
    meson
    ninja
    pkgconfig
    python36Packages.wrapPython
    wrapGAppsHook
  ];

  buildInputs = [
    gnome3.totem-pl-parser
    gnome3.libsecret
    gnome3.gnome-settings-daemon

    gst_all_1.gstreamer
    gst_all_1.gst-plugins-base

    gtk3
    libnotify
  ];

  pythonPath = with python36Packages; [
    pygobject3
    pydbus
    pycairo
  ];

  preFixup = ''
    buildPythonPath "$out/libexec/lollypop-portal $pythonPath"

    gappsWrapperArgs+=(
      --prefix PYTHONPATH : "$program_PYTHONPATH"
      --prefix PATH : "${stdenv.lib.makeBinPath [ easytag kid3 ]}"
    )
  '';

  meta = with stdenv.lib; {
    description = "DBus Service for Lollypop";
    homepage = https://gitlab.gnome.org/gnumdk/lollypop-portal;
    license = licenses.gpl3Plus;
    maintainers = with maintainers; [ worldofpeace ];
    platforms = platforms.linux;
  };
}
<commit_msg>lollypop-portal: add python3<commit_after>{ stdenv, fetchFromGitLab, meson, ninja, pkgconfig
, python3, gnome3, gst_all_1, gtk3, libnotify
, kid3, easytag, gobjectIntrospection, wrapGAppsHook }:

python3.pkgs.buildPythonApplication rec {
  name = "lollypop-portal-${version}";
  version = "0.9.7";

  format = "other";
  doCheck = false;

  src = fetchFromGitLab {
     domain = "gitlab.gnome.org";
     owner = "gnumdk";
     repo = name;
     rev = version;
     sha256 = "0rn5xmh6391i9l69y613pjad3pzdilskr2xjfcir4vpk8wprvph3";
  };

  nativeBuildInputs = [
    gobjectIntrospection
    meson
    ninja
    pkgconfig
    wrapGAppsHook
  ];

  buildInputs = [
    gnome3.gnome-settings-daemon
    gnome3.libsecret
    gnome3.totem-pl-parser
    gst_all_1.gst-plugins-base
    gst_all_1.gstreamer
    gtk3
    libnotify
    python3
  ];

  pythonPath = with python3.pkgs; [
    pycairo
    pydbus
    pygobject3
  ];

  preFixup = ''
    buildPythonPath "$out/libexec/lollypop-portal $pythonPath"
    patchPythonScript "$out/libexec/lollypop-portal"

    gappsWrapperArgs+=(
      --prefix PATH : "${stdenv.lib.makeBinPath [ easytag kid3 ]}"
    )
  '';

  meta = with stdenv.lib; {
    description = "DBus Service for Lollypop";
    homepage = https://gitlab.gnome.org/gnumdk/lollypop-portal;
    license = licenses.gpl3Plus;
    maintainers = with maintainers; [ worldofpeace ];
    platforms = platforms.linux;
  };
}

{'fraction_non_alphanumeric': 0.11023902961113093, 'fraction_numerical': 0.029611130931145202, 'mean_word_length': 3.105417276720351, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6872100', 'n_tokens_mistral': 1126, 'n_tokens_neox': 1061, 'n_words': 175}